Graphs of Derivatives

Second Book

The notebook contains the code used to create the `graphs of derivatives` on abstractmath.org. *It is raw code and contains a few errors.* Some of the graphs differ from the graphs on the website. I have no present intention to turn this into an easy-to-use package, but everyone should feel free to use and and adapt it under the CC Share-Alike license.

```
GT := 3
```

Definition of Derivative

A real valued function f may have a derivative, which is another real valued function denoted by f'. At a point a, f'[a] is the slope of the tangent line to the curve y = f[x] at the point (a, f[x]).

Functions used in creating the examples

```
ShowDerivatives[f_, n_, a_, b_] :=
 Plot[Take[{f[x], f'[x], f''[x], f'''[x], f''''[x]}, n] // Evaluate, {x, a, b},
  Prolog → AbsoluteThickness[GT], PlotRange -> {a, b}, AspectRatio -> 1,
  ImageSize → {200, 200}, PlotStyle -> {{RGBColor[0, 0, 1]}, {RGBColor[1, 0, 0]},
    {RGBColor[0, 1, 0]}, RGBColor[.7, .7, 0], RGBColor[.7, 0, .7]}]
```

TO DO : Make SD print each line in the appropriate color.

TO DO : Increase the type size in the SD printout. It is not obvious how to do this.

TO DO : Bring the graph of the function to the front so that the derivatives all pass underneath it.

```
SD[f_, n_, a_, b_, c_, d_, s_] := ((r := (d - c) / (b - a));
  Plot[Take[{f[x], f'[x], f''[x], f'''[x], f''''[x], f'''''[x]}, n] // Evaluate,
   {x, a, b}, PlotRange -> {c, d}, Prolog → AbsoluteThickness[GT],
   AspectRatio → r, ImageSize → { s, r s},
   PlotStyle -> {{RGBColor[0, 0, 1]}, {RGBColor[1, 0, 0]}, {RGBColor[0, 1, 0]},
     RGBColor[.7, .7, 0], RGBColor[.7, 0, .7], RGBColor[0, .7, .7]}])
```

The derivatives are shown in each graph by color.
function: blue

first derivative : red
second : green
third : gold
fourth : purple
fifth: light blue

DF prints out the formulas for the derivatives

```
DF[f_, n_] :=
 TraditionalForm[
  TableForm[
   Take[{{"function", f[x]}, {"1st deriv", f'[x]},
     {"2nd deriv", f''[x]}, {"3rd deriv", f'''[x]}, {"4th deriv", f''''[x]},
     {"5th deriv", f'''''[x]}, {"6th deriv", f''''''[x]}}, n]]]
```

Transcendental functions

See `transcendental functions` in Wikipedia.

Sine

```
DSolve[u[x] == u''''[x], u[x], x]
```

{{u[x] → e^x C[1] + e^{-x} C[3] + C[2] Cos[x] + C[4] Sin[x]}}

The sine is its own fourth derivative.

```
DF[Sin, 4]
```

function	$\sin(x)$
1st deriv	$\cos(x)$
2nd deriv	$-\sin(x)$
3rd deriv	$-\cos(x)$

```
SD[Sin, 2, -2 Pi, 2 Pi, -1.1, 1.1, 400]
```




```
SD[Sin, 4, -2 Pi, 2 Pi, -1.1, 1.1, 400]
```



```
Manipulate[
 SD[Sin[a # + b] &, 4, -2 π, 2 π, -3.6, 3.6, 400],
 {{a, 1}, -1.5, 1.5, Appearance → "Open"},
 {{b, 0}, -2 π, 2 π, Appearance → "Open"},
 SaveDefinitions → True]
```



■ Cosine

The cosine is its own fourth derivative.

```
DF[Cos, 4]
```

function	$\cos(x)$
1st deriv	$-\sin(x)$
2nd deriv	$-\cos(x)$
3rd deriv	$\sin(x)$

```
SD[Cos, 4, -2 Pi, 2 Pi, -1.1, 1.1, 400]
```



Sine + Cosine

```
splusc[x_] := Sin[x] + Cos[x]
```

```
DF[splusc, 4]
```

function	$\sin(x) + \cos(x)$
1st deriv	$\cos(x) - \sin(x)$
2nd deriv	$-\sin(x) - \cos(x)$
3rd deriv	$\sin(x) - \cos(x)$

```
SD[splusc, 4, -2 Pi, 2 Pi, -2.1, 2.1, 400]
```



Sine times Cosine

Note how the amplitude gets bigger with each derivative

```
sxc[x_] := Sin[x] Cos[x]
```

```
DF[sxc, 4]
```

function	$\sin(x)\cos(x)$
1st deriv	$\cos^2(x) - \sin^2(x)$
2nd deriv	$-4\sin(x)\cos(x)$
3rd deriv	$4\sin^2(x) - 4\cos^2(x)$

```
SD[sxc, 4, -2 Pi, 2 Pi, -4.1, 4.1, 400]
```



Tangent

```
Pi / 2 // N
```

1.5708

The derivatives all contain an occurrence of Tangent, and since the cos(π/2) is zero, the function and all its derivatives have asymptotes at odd multiples of π/2. The graph shows four of the asymptotes (vertical lines).

```
DF[Tan, 5]
```

function	$\tan(x)$
1st deriv	$\sec^2(x)$
2nd deriv	$2\tan(x)\sec^2(x)$
3rd deriv	$2\sec^4(x) + 4\tan^2(x)\sec^2(x)$
4th deriv	$16\tan(x)\sec^4(x) + 8\tan^3(x)\sec^2(x)$

```
SD[Tan, 5, -4.9, 4.9, -3, 4, 400]
```




```
tanover2[x_] := Tan[x / 2]
```

```
DF[tanover2, 5]
```

function	$\tan\left(\frac{x}{2}\right)$
1st deriv	$\frac{1}{2}\sec^2\left(\frac{x}{2}\right)$
2nd deriv	$\frac{1}{2}\tan\left(\frac{x}{2}\right)\sec^2\left(\frac{x}{2}\right)$
3rd deriv	$\frac{1}{4}\sec^4\left(\frac{x}{2}\right) + \frac{1}{2}\tan^2\left(\frac{x}{2}\right)\sec^2\left(\frac{x}{2}\right)$
4th deriv	$\tan\left(\frac{x}{2}\right)\sec^4\left(\frac{x}{2}\right) + \frac{1}{2}\tan^3\left(\frac{x}{2}\right)\sec^2\left(\frac{x}{2}\right)$

```
SD[tanover2, 5, -Pi, Pi, -4, 4, 250]
```



Secant

The vertical lines are the asymptotes.

```
secant[x_] := 1/Cos[x]
```

```
DF[secant, 5]
```

function	$\sec(x)$
1st deriv	$\tan(x)\sec(x)$
2nd deriv	$\sec^3(x) + \tan^2(x)\sec(x)$
3rd deriv	$5\tan(x)\sec^3(x) + \tan^3(x)\sec(x)$
4th deriv	$5\sec^5(x) + 18\tan^2(x)\sec^3(x) + \tan^4(x)\sec(x)$

```
SD[secant, 5, -4.9, 4.9, -4, 4, 400]
```




```
secsq[x_] := 1 / (Cos[x]) ^2
```

ArcTangent

The number of critical points is 0 for the function, 1 for the first derivative, 2 for the second derivative, and increments that way at least thru the 5 th derivative. Exercise : Prove it always happens!

```
DF[ArcTan, 5]
```

function	$\tan^{-1}(x)$
1st deriv	$\frac{1}{x^2+1}$
2nd deriv	$-\frac{2x}{(x^2+1)^2}$
3rd deriv	$\frac{8x^2}{(x^2+1)^3} - \frac{2}{(x^2+1)^2}$
4th deriv	$\frac{24x}{(x^2+1)^3} - \frac{48x^3}{(x^2+1)^4}$

```
SD[ArcTan, 5, -4, 4, -5, 5, 450]
```



Exponential

the exponential function

The exponential function is its own derivative

```
DF[Exp, 6]
```

function	e^x
1st deriv	e^x
2nd deriv	e^x
3rd deriv	e^x
4th deriv	e^x
5th deriv	e^x

```
SD[Exp, 1, -2, 2, .1, 6, 200]
```



modified slightly to spread the derivatives out

```
expmod1[x_] := Exp[.8 x]
```

```
DF[expmod1, 6]
```

function	$e^{0.8\,x}$
1st deriv	$0.8\,e^{0.8\,x}$
2nd deriv	$0.64\,e^{0.8\,x}$
3rd deriv	$0.512\,e^{0.8\,x}$
4th deriv	$0.4096\,e^{0.8\,x}$
5th deriv	$0.32768\,e^{0.8\,x}$

```
SD[expmod1, 6, -2, 2, .1, 5, 200]
```



```
Manipulate[
 SD[Exp[a #] &, 6, -2, 2, -5, 5, 100],
 {{a, 1}, -1.5, 1.5, Appearance → "Open"},
 SaveDefinitions → True]
```



A function that is its own second derivative

```
DSolve[u''[x] == u[x], u[x], x]
```

{{u[x] → e^x C[1] + e^{-x} C[2]}}

```
ownsd[x_] := Exp[x] + Exp[-x]
```

```
DF[ownsd, 2]
```

function	$e^{-x} + e^x$
1st deriv	$e^x - e^{-x}$

```
SD[ownsd, 2, -2, 2, -2, 5, 200]
```



```
Manipulate[
 SD[Exp[-a #] + Exp[b #] &, 6, -2, 2, -5, 5, 100],
 {{a, 1}, -1.5, 1.5, Appearance → "Open"},
 {{b, 1}, -1.5, 1.5, Appearance → "Open"},
 SaveDefinitions → True]
```



A function that is its own third derivative

TO DO : This function has very complicated behavior left of 0 that I don' t understand because it takes on extreme values that I cannot plot.

```
DSolve[u'''[x] == u[x], u[x], x]
```

$$\left\{\left\{u[x] \to e^{x} C[1] + e^{-x/2} C[2] \operatorname{Cos}\left[\frac{\sqrt{3}\ x}{2}\right] + e^{-x/2} C[3] \operatorname{Sin}\left[\frac{\sqrt{3}\ x}{2}\right]\right\}\right\}$$

$$\text{owntd}[x\_] := e^{x} + e^{-x/2} \operatorname{Cos}\left[\frac{\sqrt{3}\ x}{2}\right]$$

```
Manipulate[owntd[x], {x, -50, 2, Appearance → "Open"}] // N
```




```
DF[owntd, 4]
```

function	$e^x + e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right)$
1st deriv	$e^x - \frac{1}{2}\sqrt{3}\,e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right) - \frac{1}{2}e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right)$
2nd deriv	$e^x + \frac{1}{2}\sqrt{3}\,e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right) - \frac{1}{2}e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right)$
3rd deriv	$e^x + e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right)$

```
SD[owntd, 3, -6.4, 4, -7.5, 13.2, 200]
```



```
SD[owntd, 3, -50, 50, -70, 50, 300]
```



Log

Log

```
Log[x]
```

Log[x]

```
DF[Log, 6]
```

function	$\log(x)$
1st deriv	$\frac{1}{x}$
2nd deriv	$-\frac{1}{x^2}$
3rd deriv	$\frac{2}{x^3}$
4th deriv	$-\frac{6}{x^4}$
5th deriv	$\frac{24}{x^5}$

```
GT := 2
```

```
SD[Log, 6, 0.01, 11, -4, 4, 350]
```



Log to base 10

```
log10[x_] := Log[10, x]
```

```
DF[log10, 6]
```

function	$\frac{\log(x)}{\log(10)}$
1st deriv	$\frac{1}{x\log(10)}$
2nd deriv	$-\frac{1}{x^2\log(10)}$
3rd deriv	$\frac{2}{x^3\log(10)}$
4th deriv	$-\frac{6}{x^4\log(10)}$
5th deriv	$\frac{24}{x^5\log(10)}$

```
SD[log10, 6, 0.01, 11, -4, 4, 400]
```



x times log

```
xxlog[x_] := x Log[x]
```

```
DF[xxlog, 5]
```

function	$x \log(x)$
1st deriv	$\log(x) + 1$
2nd deriv	$\frac{1}{x}$
3rd deriv	$-\frac{1}{x^2}$
4th deriv	$\frac{2}{x^3}$

```
SD[xxlog, 5, 0.01, 5, -4, 4, 300]
```



Log over x

```
logoverx[x_] := Log[x] / x
```

```
DF[logoverx, 6]
```

function	$\frac{\log(x)}{x}$
1st deriv	$\frac{1}{x^2} - \frac{\log(x)}{x^2}$
2nd deriv	$\frac{2\log(x)}{x^3} - \frac{3}{x^3}$
3rd deriv	$\frac{11}{x^4} - \frac{6\log(x)}{x^4}$
4th deriv	$\frac{24\log(x)}{x^5} - \frac{50}{x^5}$
5th deriv	$\frac{274}{x^6} - \frac{120\log(x)}{x^6}$

```
SD[logoverx, 6, 0.01, 4, -3, 4, 300]
```



Miscellaneous combinations

Sum of a polynomial and Sine

```
spsine[x_] := 2 + (1 / 30) (x - 3) ^2 + Sin[ x]
```

```
DF[spsine, 5]
```

function	$\frac{1}{30}(x-3)^2+\sin(x)+2$
1st deriv	$\frac{x-3}{15}+\cos(x)$
2nd deriv	$\frac{1}{15}-\sin(x)$
3rd deriv	$-\cos(x)$
4th deriv	$\sin(x)$

Note that the first derivative (red) is rising slowly.

```
SD[spsine, 5, -20, 20, -3, 20, 500]
```



Rational times Sin

RatSin1

```
ratsin1[x_] := (4 / (x - 1) ) Sin[x]
```

```
DF[ratsin1, 5]
```

function	$\frac{4\sin(x)}{x-1}$
1st deriv	$\frac{4\cos(x)}{x-1} - \frac{4\sin(x)}{(x-1)^2}$
2nd deriv	$-\frac{4\sin(x)}{x-1} + \frac{8\sin(x)}{(x-1)^3} - \frac{8\cos(x)}{(x-1)^2}$
3rd deriv	$\frac{12\sin(x)}{(x-1)^2} - \frac{24\sin(x)}{(x-1)^4} - \frac{4\cos(x)}{x-1} + \frac{24\cos(x)}{(x-1)^3}$
4th deriv	$\frac{4\sin(x)}{x-1} - \frac{48\sin(x)}{(x-1)^3} + \frac{96\sin(x)}{(x-1)^5} + \frac{16\cos(x)}{(x-1)^2} - \frac{96\cos(x)}{(x-1)^4}$

The vertical purple line is an asymptote

```
SD[ratsin1, 5, -4 Pi, 4 Pi, -7, 7, 600]
```



RatSin2

This one, unlike the one above, has no asymptote. What gives?? Answer: At 0, the value of the function has the form 0/0 so it is undetermined. That means the point (0,4) *is not on the graph of the function*. (Similar statements are true of the derivatives). Of course, you can't see the hole! .

If you define Ratsin2[0] := 4 the result is that ratsin2 is continuous at 0, as both the graph and the table below suggest

```
ratsin2[x_] := (4 / x ) Sin[x]
```

```
ratsin2[0]
```

Power::infy : Infinite expression $\frac{1}{0}$ encountered. >>

Infinity::indet : Indeterminate expression 0 ComplexInfinity encountered. >>

```
Indeterminate
```

```
Table[{n/10, ratsin2[n/10]}, {n, {-10, -9, -7, -6, -5,
    -4, -3, -2, -1, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10}}] // N // TableForm
```

-1.	3.36588
-0.9	3.48145
-0.7	3.68124
-0.6	3.76428
-0.5	3.8354
-0.4	3.89418
-0.3	3.94027
-0.2	3.97339
-0.1	3.99334
0.1	3.99334
0.2	3.97339
0.3	3.94027
0.4	3.89418
0.5	3.8354
0.6	3.76428
0.7	3.68124
0.8	3.58678
0.9	3.48145
1.	3.36588

```
{ratsin2[-1/10], ratsin[0], ratsin[1/10]}
```

```
{40 Sin[1/10], ratsin[0], ratsin[1/10]}
```

```
DF[ratsin2, 5]
```

function	$\frac{4\sin(x)}{x}$
1st deriv	$\frac{4\cos(x)}{x} - \frac{4\sin(x)}{x^2}$
2nd deriv	$\frac{8\sin(x)}{x^3} - \frac{8\cos(x)}{x^2} - \frac{4\sin(x)}{x}$
3rd deriv	$-\frac{24\sin(x)}{x^4} + \frac{24\cos(x)}{x^3} + \frac{12\sin(x)}{x^2} - \frac{4\cos(x)}{x}$
4th deriv	$\frac{96\sin(x)}{x^5} - \frac{96\cos(x)}{x^4} - \frac{48\sin(x)}{x^3} + \frac{16\cos(x)}{x^2} + \frac{4\sin(x)}{x}$

```
SD[ratsin2, 5, -4 Pi, 4 Pi, -7, 7, 600]
```



RatSin3

No asymptote problem here. The denominator of the fraction is never 0.

```
ratsin3[x_] := (3 / (x^2 + 1)) Sin[x]
```

```
DF[ratsin3, 5]
```

function $\frac{3\sin(x)}{x^2+1}$

1st deriv $\frac{3\cos(x)}{x^2+1} - \frac{6x\sin(x)}{(x^2+1)^2}$

2nd deriv $\frac{24x^2\sin(x)}{(x^2+1)^3} - \frac{3\sin(x)}{x^2+1} - \frac{6\sin(x)}{(x^2+1)^2} - \frac{12x\cos(x)}{(x^2+1)^2}$

3rd deriv $\frac{18x\sin(x)}{(x^2+1)^2} + \frac{72x\sin(x)}{(x^2+1)^3} + \frac{72x^2\cos(x)}{(x^2+1)^3} - \frac{3\cos(x)}{x^2+1} - \frac{18\cos(x)}{(x^2+1)^2} - \frac{144x^3\sin(x)}{(x^2+1)^4}$

4th deriv $-\frac{144x^2\sin(x)}{(x^2+1)^3} - \frac{864x^2\sin(x)}{(x^2+1)^4} + \frac{3\sin(x)}{x^2+1} + \frac{36\sin(x)}{(x^2+1)^2} + \frac{72\sin(x)}{(x^2+1)^3} + \frac{24x\cos(x)}{(x^2+1)^2} + \frac{288x\cos(x)}{(x^2+1)^3} + \frac{1152x^4\sin(x)}{(x^2+1)^5} - \frac{576x^3\cos(x)}{(x^2+1)^4}$

```
SD[ratsin3, 3, -2 Pi, 2 Pi, -5.1, 5.1, 400]
```



Sine of a polynomial

```
f[x_] := Sin[x^2 - 1]
```

```
ShowDerivatives[f, 2, -2, 2]
```




```
g[x_] := 2 x
```

```
Plot[{f[x], f'[x], g[x]}, {x, -3, 3}, PlotRange → {-6, 6},
 ImageSize → {200, 2 × 200}, AspectRatio → 2, Prolog → AbsoluteThickness[GT],
 PlotStyle -> {RGBColor[0, 0, 1], RGBColor[1, 0, 0], RGBColor[0, 0, 0]}]
```



```
Plot[{f[x], f'[x], g[x]}, {x, -4, 4}, PlotRange → {-8, 8},
 ImageSize → {300, 2 × 300}, AspectRatio → 2, Prolog → AbsoluteThickness[GT],
 PlotStyle -> {RGBColor[0, 0, 1], RGBColor[1, 0, 0], RGBColor[0, 0, 0]}]
```



```
Plot[{f[x], f'[x]}, {x, -4, 4}, PlotRange → {-9, 9},
 ImageSize → {200, 2 × 200}, AspectRatio → 2.25, Prolog → AbsoluteThickness[GT],
 PlotStyle -> {RGBColor[0, 0, 1], RGBColor[1, 0, 0], RGBColor[0, 1, 0]}]
```



ArcSin

```
f[x_] := ArcSin[x]
```

```
DF[f, 5]
```

function	$\sin^{-1}(x)$
1st deriv	$\frac{1}{\sqrt{1-x^2}}$
2nd deriv	$\frac{x}{(1-x^2)^{3/2}}$
3rd deriv	$\frac{3x^2}{(1-x^2)^{5/2}} + \frac{1}{(1-x^2)^{3/2}}$
4th deriv	$\frac{9x}{(1-x^2)^{5/2}} + \frac{15x^3}{(1-x^2)^{7/2}}$

```
SD[f, 4, -1, 1, -2, 3, 200]
```




```
Remove[f]
```

ArcSec

```
f[x_] := ArcSec[x]
```

```
DF[f, 5]
```

function	$\sec^{-1}(x)$
1st deriv	$\frac{1}{\sqrt{1-\frac{1}{x^2}}\, x^2}$
2nd deriv	$-\frac{1}{x^5\left(1-\frac{1}{x^2}\right)^{3/2}}-\frac{2}{x^3\sqrt{1-\frac{1}{x^2}}}$
3rd deriv	$\frac{3}{x^8\left(1-\frac{1}{x^2}\right)^{5/2}}+\frac{7}{x^6\left(1-\frac{1}{x^2}\right)^{3/2}}+\frac{6}{x^4\sqrt{1-\frac{1}{x^2}}}$
4th deriv	$-\frac{15}{x^{11}\left(1-\frac{1}{x^2}\right)^{7/2}}-\frac{45}{x^9\left(1-\frac{1}{x^2}\right)^{5/2}}-\frac{48}{x^7\left(1-\frac{1}{x^2}\right)^{3/2}}-\frac{24}{x^5\sqrt{1-\frac{1}{x^2}}}$

```
SD[f, 5, 1, 5, -3, 3, 300]
```




```
Remove[f]
```

■ x times sin

```
f[x_] := x Sin[x]
```

```
ShowDerivatives[f, 5, -7, 7]
```